Instructor: Alessandra Buonanno (`alessandra.buonanno@hu-berlin.de`)
Tutor *(corresponding for this sheet)*: Aldo Gamboa (`aldo.gamboa@aei.mpg.de`)
Tutor: Marcus Haberland (`marcus.haberland@aei.mpg.de`)

1. **On the effective-one-body Hamiltonian and dynamics**

We start from Eq (8) of the assignment, relating the real and effective Hamiltonians

$$1+\hat{H}(q,p)\varepsilon^2\left(1+\alpha_1\varepsilon^2\hat{H}(q,p)\right)=\hat{H}_{\text{eff}}(Q,P)\varepsilon^2 \tag{1}$$

where for simplicity we call $\hat{H}_{\text{real}}=\hat{H}$, and we remind that

$$\hat{H}(q,p)=\hat{H}_{\text{Newt}}(q,p)+\varepsilon^2\hat{H}_{\text{1PN}}(q,p)+\cdots \tag{2}$$

$$\hat{H}_{\text{Newt}}\;(q,p)=\frac{1}{2}p^2-\frac{1}{q}, \tag{3}$$

$$\hat{H}_{\text{1PN}}(q,p)=-\frac{1}{8}(1-3\nu)p^4-\frac{1}{2q}\left[(3+\nu)p^2+\nu(\mathbf{n}\cdot\mathbf{p})^2\right]+\frac{1}{2q^2}. \tag{4}$$

The canonical transformation at 1PN order has generating function

$$G=q^ip_i\left[c_1p^ip_i+\frac{c_2}{\sqrt{q^iq_i}}\right] \tag{5}$$

whose derivatives are given by

$$\partial_{p_i}G=q_i\left[c_1p^2+\frac{c_2}{q}\right]+q^jp_j\left(c_12p_i\right) \tag{6}$$

$$\partial_{q_i}G=p^i\left[c_1p^2+\frac{c_2}{q}\right]-q^jp_j\left(c_2\frac{1}{q^3}q^i\right). \tag{7}$$

From the 1PN canonical transformation

$$Q^i=q^i+\frac{1}{c^2}\frac{\partial G_{\text{1PN}}}{\partial p_i}, \tag{8}$$

$$P_i=p_i-\frac{1}{c^2}\frac{\partial G_{\text{1PN}}}{\partial q^i} \tag{9}$$

we can find Q,P,N_iP_i at 1PN (note that N_iP_i will not actually be needed).

$$\begin{aligned}Q=\sqrt{Q^iQ_i}&=\sqrt{q^iq_i+2q^i\partial_{p_i}G\varepsilon^2}\\&=\sqrt{q^iq_i+2\varepsilon^2\left[q^2\left(c_1p^2+\frac{c_2}{q}\right)+2(\mathbf{q}\cdot\mathbf{p})^2c_1\right]}\end{aligned} \tag{10}$$

$$\begin{aligned}P=\sqrt{p^ip_i}&=\sqrt{p^ip_i-2p^i\partial_{q_i}G\varepsilon^2}\\&=\sqrt{p^ip_i-2\varepsilon^2\left[p^2\left(c_1p^2+\frac{c_2}{q}\right)-(\mathbf{q}\cdot\mathbf{p})^2\frac{c_2}{q^3}\right]}\end{aligned} \tag{11}$$

$$\begin{aligned} N\cdot P &= \frac{Q^i}{Q}P_i = \frac{1}{Q}\left(\mathbf{p}\cdot\mathbf{q} + \varepsilon^2\left(p^i\partial_{p_i}G - q^i\partial_{q_i}G\right)\right) \\ &= \frac{1}{Q}\left(\mathbf{p}\cdot\mathbf{q} + \varepsilon^2\left(\mathbf{p}\cdot\mathbf{q}\left(\cancel{c_1p^2} + \frac{c_2}{q} + 2c_1p^2\right) - \mathbf{p}\cdot\mathbf{q}\left(\cancel{c_1p^2} + \cancel{\frac{c_2}{q}} - \cancel{\frac{c_2}{q}}\right)\right)\right) \\ &= \frac{1}{Q}\left(\mathbf{p}\cdot\mathbf{q} + \varepsilon^2\left[\mathbf{p}\cdot\mathbf{q}\left(2c_1p^2 + \frac{c_2}{q}\right)\right]\right) \qquad \text{(expand } Q\text{)} \\ &\simeq \frac{1}{q}\left(\mathbf{p}\cdot\mathbf{q} + \varepsilon^2\left[\mathbf{p}\cdot\mathbf{q}\left(2c_1p^2 + \frac{c_2}{q}\right)\right]\right)\left(1 + 2\varepsilon^2\left[c_1p^2 + \frac{c_2}{q} + 2\left(\frac{\mathbf{q}\cdot\mathbf{p}}{q}\right)^2 c_1\right]\right) \end{aligned} \tag{12}$$

Let's match the square of the Hamiltonian order by order:

$$\left[1 + \varepsilon^2\hat{H}\left(1 + \alpha_1\varepsilon^2\hat{H}\right)\right]^2 = \hat{H}^2_{\text{eff}}\,\varepsilon^4$$

$$1 + 2\varepsilon^2\hat{H}\left(1 + \alpha_1\varepsilon^2\hat{H}\right) + \varepsilon^4\hat{H}^2\left(1 + \alpha_1\varepsilon^2\hat{H}\right)^2 = \hat{H}^2_{\text{eff}}\,\varepsilon^4 \tag{13}$$

Let's consider first the right hand side:

$$\begin{aligned} \varepsilon^4\hat{H}^2_{\text{eff}} &= \left[\left(1 + \frac{a_1}{Q}\varepsilon^2\right)\left(1 + \varepsilon^2P^2 + \varepsilon^4 a_1\frac{(\mathbf{N}\cdot\mathbf{P})^2}{Q}\right)\right] \\ &= 1 + \varepsilon^2P^2 + \varepsilon^4a_1\frac{(\mathbf{N}\cdot\mathbf{P})^2}{Q} + \frac{a_1}{Q}\varepsilon^2 + \frac{a_1P^2}{Q}\varepsilon^4 + \mathcal{O}\left(\varepsilon^6\right) \\ &= 1 + \varepsilon^2\left(P^2 + \frac{a_1}{Q}\right) + \varepsilon^4\left(a_1\frac{(\mathbf{N}\cdot\mathbf{P})^2}{Q} + \frac{a_1P^2}{Q}\right) + \mathcal{O}\left(\varepsilon^6\right) \end{aligned} \tag{14}$$

Left hand side:

$$\begin{aligned} &1 + 2\varepsilon^2\hat{H}\left(1 + \alpha_1\varepsilon^2\hat{H}\right) + \varepsilon^4H^2\left(1 + \alpha_1\varepsilon^2\hat{H}\right)^2 \\ &1 + 2\varepsilon^2\hat{H} + 2\alpha_1\hat{H}^2\varepsilon^4 + \varepsilon^4\hat{H}^2 + \mathcal{O}\left(\varepsilon^6\right) \\ &1 + 2\varepsilon^2\hat{H} + \varepsilon^4\hat{H}^2\left(2\alpha_1 + 1\right) + \mathcal{O}\left(\varepsilon^6\right) \qquad \left(\hat{H} = \hat{H}_N + \varepsilon^2H_{1\text{PN}}\right) \\ &1 + 2\varepsilon^2\hat{H}_N + \varepsilon^4\left(2\hat{H}_{1\text{PN}} + \hat{H}_N^2\left(2\alpha_1 + 1\right)\right) + \mathcal{O}\left(\varepsilon^6\right) \end{aligned} \tag{15}$$

Matching at $\mathcal{O}(\varepsilon^2)$:

$$2\hat{H}_N \overset{!}{=} P^2 + \frac{a_1}{Q}$$

$$\begin{aligned} p^2 - \frac{2}{q} \overset{!}{=}\, & p^ip_i - 2\varepsilon^2\left[p^2\left(c_1p^2 + \frac{c_2}{q}\right) - (\mathbf{q}\cdot\mathbf{p})^2\frac{c_2}{q^3}\right] + \\ & + a_1\left(q^iq_i + 2\varepsilon^2\left[q^2\left(c_1\mathbf{p}^2 + \frac{c_2}{q}\right) + 2c_1(\mathbf{q}\cdot\mathbf{p})^2\right]\right)^{-\frac{1}{2}} \\ & = p^2 - 2\varepsilon^2\left[p^2\left(c_1p^2 + \frac{c_2}{q}\right) - (\mathbf{q}\cdot\mathbf{p})^2\frac{c_2}{q^3}\right]\Bigg) \\ & + \frac{a_1}{q}\left(1 - \varepsilon^2\left[\cancel{\frac{q^2}{q^2}}\left(c_1p^2 + \frac{c_2}{q}\right) + \frac{2c_1(\mathbf{q}\cdot\mathbf{p})^2}{q^2}\right]\right) \end{aligned} \tag{16}$$

From which we obtain $a_1 = -2$. The $\mathcal{O}(\varepsilon^2)$ terms will also be relevant at the next order. Matching at $\mathcal{O}(\varepsilon^4)$, the LHS is given by:

$$2\left(-\frac{1-3\nu}{8}p^4 - \frac{1}{2q}\left[(3+\nu)p^2 + \nu(\mathbf{n}\cdot\mathbf{p})^2\right] + \frac{1}{2q^2}\right) + (2\alpha_1+1)\left(\frac{p^4}{4} + \frac{1}{q^2} - \frac{p^2}{q}\right) \tag{17}$$

RHS:

$$\overbrace{\left[a_1\frac{(\mathbf{N}\cdot\mathbf{P})^2}{Q} + \frac{a_1 P^2}{Q}\right]}^{\text{To be considered at}\mathcal{O}(\varepsilon^0)} + \left(\frac{-a_1}{q}\right)\left[\left(c_1p^2 + \frac{c_2}{q}\right) + \frac{2c_1(\mathbf{q}\cdot\mathbf{p})^2}{q^2}\right] - 2\overbrace{\left[p^2\left(c_1p^2 + \frac{c_2}{q}\right) - (\mathbf{q}\cdot\mathbf{p})^2\frac{c_2}{q^3}\right]}^{\text{From previous order terms}} \tag{18}$$

RHS, expanding and putting terms together:

$$\begin{aligned} &a_1\left(\frac{(\mathbf{n}\cdot\mathbf{p})^2}{q} + \frac{p^2}{q} - \frac{p^2c_1}{q} - \frac{c_2}{q^2} - 2c_1\frac{(\mathbf{q}\cdot\mathbf{p})^2}{q^3}\right) - 2\left(c_1p^4 + c_2\frac{p^2}{q} - \frac{(\mathbf{q}\cdot\mathbf{p})^2}{q^3}c_2\right) = \\ &\left[\frac{(\mathbf{n}\cdot\mathbf{p})^2}{q}\left(-2+2c_2-2\left(-2c_1\right)\right)\right] + \left[\frac{p^2}{q}\left(-2+2c_1-2c_2\right)\right] + \left[2\frac{c_2}{q^2} - 2c_1p^4\right] \end{aligned} \tag{19}$$

LHS:

$$\begin{aligned} &p^4\left[-\frac{1-3\nu}{4} + \frac{2\alpha_1+1}{4}\right] + \left[2\alpha_1+1+1\right]\frac{1}{q^2} \\ &+\left[-\left(2\alpha_1+1\right) - 3 - \nu\right]\frac{p^2}{q} + \left[-\nu\right]\frac{(\mathbf{n}\cdot\mathbf{p})^2}{q} \end{aligned} \tag{20}$$

By matching the coefficients of the corresponding terms, and remembering that $\frac{(\mathbf{n}\cdot\mathbf{p})^2}{q} = \frac{(\mathbf{q}\cdot\mathbf{p})^2}{q^3}$, we obtain the following equations:

$$c_2 = \alpha_1 + 1 \tag{21}$$

$$-2c_1 = \frac{2\alpha_1+1-1+3\nu}{4} = \frac{2\alpha_1+3\nu}{4} \implies c_1 = -\frac{2\alpha_1+3\nu}{8} \tag{22}$$

$$-2+2c_2-2\left(-2c_1\right) = -\nu \rightarrow +4\alpha_1 - 2\alpha_1 - 3\nu = -2\nu \implies \alpha_1 = \frac{\nu}{2} \tag{23}$$

$$c_1 = -\frac{\nu}{2}, \quad c_2 = 1 + \frac{\nu}{2} \tag{24}$$

2. **Incorporating the emission of gravitational waves in the two-body dynamics:**

a) For the energy-flux relation, we calculate the total time derivative of the Hamiltonian with the chain rule:

$$\frac{dH}{dt} = \frac{\partial H}{\partial t} + \frac{\partial H}{\partial r}\dot{r} + \frac{\partial H}{\partial \phi}\dot{\phi} + \frac{\partial H}{\partial p_r}\dot{p}_r + \frac{\partial H}{\partial p_\phi}\dot{p}_\phi. \tag{25}$$

If the Hamiltonian does not depend explicitly on time or in the azimuthal angle ϕ, we can simplify this to

$$\frac{dH}{dt} = \frac{\partial H}{\partial r}\dot{r} + \frac{\partial H}{\partial p_r}\dot{p}_r + \frac{\partial H}{\partial p_\phi}\dot{p}_\phi. \tag{26}$$

Next, we substitute Hamilton's EOMs to get:

$$\frac{dH}{dt} = \frac{\partial H}{\partial r}\frac{\partial H}{\partial p_r} + \frac{\partial H}{\partial p_r}\left(-\frac{\partial H}{\partial r} + \mathcal{F}_r\right) + \frac{\partial H}{\partial p_\phi}\mathcal{F}_\phi \tag{27}$$

$$= \dot{r}\mathcal{F}_r + \dot{\phi}\mathcal{F}_\phi, \tag{28}$$

which is the desired result since $dH/dt = -\Phi_E$.
The angular-momentum-flux relation is directly obtained by balancing the change in the angular momentum (given by Hamilton's EOMs) with the angular momentum flux:

$$\dot{p}_\phi = -\Phi_L = \mathcal{F}_\phi. \tag{29}$$

b) To compute the orbit averages, we would need to obtain some relations within the Keplerian parametrization. First, we note that the orbit average of a function f over an orbital period can be calculated as

$$\langle f\rangle \equiv \frac{1}{T}\oint f\,dt = \frac{1}{T}\int_0^{2\pi}\frac{f}{\dot{\phi}}d\phi, \tag{30}$$

where $T = \oint dt = \int_0^{2\pi} 1/\dot{\phi}\,d\phi$. To compute these integrals, we employ the relation between the angular momentum and the angular velocity of the orbit:

$$\dot{\phi} = \frac{L}{\mu r^2}, \tag{31}$$

where

$$r = \frac{R}{1+e\cos\phi} = \frac{a(1-e^2)}{1+e\cos\phi}. \tag{32}$$

We will also need the following relations from the Kepler's problem:

$$R = \frac{L^2}{GM\mu^2}, \tag{33}$$

$$e = 1 + \frac{2EL^2}{GM^2\mu^3}, \tag{34}$$

where E is the energy of the orbit.
Employing these relations, we first get

$$T = 2\pi\sqrt{\frac{a^3}{GM}}. \tag{35}$$

Now, we move on to calculate the quadrupole moment Q_{ij} of the source:

$$Q_{ij} = M_{ij} - \frac{1}{3}\delta_{ij}\mathrm{Tr}\,(M), \tag{36}$$

$$M_{ij} = \mu x_i x_j = \mu r^2\begin{pmatrix}\cos^2\phi & \sin\phi\cos\phi & 0\\ \sin\phi\cos\phi & \sin^2\phi & 0\\ 0 & 0 & 1\end{pmatrix}, \tag{37}$$

where we have used a polar parametrization of the orbit, such that $x = r\cos\phi$ and $y = r\sin\phi$.

To take the time derivatives of the quadrupole, we can use three times (for the three time derivatives) the relation

$$\dot{Q}_{ij} = \frac{dQ_{ij}}{d\phi}\dot{\phi}, \tag{38}$$

where the r inside Q_{ij} is expressed via Eq. (32), and in this way, we will get expressions as functions of ϕ, which can be integrated to get the orbit-averages.

An alternative (more generic) approach is to take the time derivatives of the multipole moments *without* specifying the Keplerian parametrization of the orbit, and then, every time a time derivative appears ($\dot{r}, \dot{\phi}, \dot{p}_r, \dot{p}_\phi$), we substitute Hamilton's EOMs for the Kepler problem. In the attached `Mathematica` solution, we follow the latter strategy, and we work in reduced variables.

The solution, in physical units, is found to be

$$\langle \Phi_E \rangle = \frac{32}{5}\frac{G^4\mu^2 M^3}{c^5 a^5}\frac{1}{(1-e^2)^{7/2}}\left(1 + \frac{73}{24}e^2 + \frac{37}{96}e^4\right), \tag{39a}$$

$$\langle \Phi_L \rangle = \frac{32}{5}\frac{G^{7/2}\mu^2 M^{5/2}}{c^5 a^{7/2}}\frac{1}{(1-e^2)^2}\left(1 + \frac{7}{8}e^2\right). \tag{39b}$$

c) Since energy is conserved along the orbit, then we evaluate the Hamiltonian at the two turning points $\phi = 0, \pi$ where $p_r = 0$, and then we can solve the following system of equations in terms of p_ϕ:

$$E = \frac{p_\phi^2}{2r_+^2} - \frac{1}{r_+}, \tag{40}$$

$$E = \frac{p_\phi^2}{2r_-^2} - \frac{1}{r_-}, \tag{41}$$

where

$$r_+ = \frac{R}{1+e} \qquad \text{and} \qquad r_- = \frac{R}{1-e}. \tag{42}$$

Doing this, and employing the relation $R = a(1-e^2)$ we find

$$p_\phi = L = \sqrt{a(1-e^2)}. \tag{43}$$

Evaluating this relation in Eq. (40), we get

$$E = -\frac{1}{2a}. \tag{44}$$

Finally, using Eqs. (43) and (44) in the generic expression of the Hamiltonian, we get

$$p_r = \frac{e\sin\phi}{\sqrt{a(1-e^2)}}. \tag{45}$$

d) This problem is solved by direct integration of the involved quantities. We require Hamilton's EOM for $\dot{\phi}$ and $\dot{r}$, as well as the expressions for r, p_ϕ, and p_r given by Eqs. (32), (43), and (45), respectively. The orbit average is computed with the relation given in Eq. (30). Since this is a straightforward process, we do it in the attached `Mathematica` notebook, where it is shown that we recover the expected results.

e) This problem is also treated in the attached `Mathematica` notebook. Below is shown a plot of the trajectory.



FIG. 1. Trajectory in the xy-plane for an equal-mass binary with starting values of $e = 0.3$, $R = 20$, $\phi = 0$, $\alpha = -16/3$, and $\beta = -13/2$.

Solution to HW4

GW course 2024

Prof. Alessandra Buonanno
TAs: Aldo Gamboa and Marcus Haberland

In[1]:=
```
$Assumptions = {a > 0, e > 0, e < 1};
```

2b)

In[2]:=
```
addtdependence = {ϕ → ϕ[t], r → r[t], pr → pr[t], pϕ → pϕ[t]};
removetdependence = {ϕ[t] → ϕ, r[t] → r, pr[t] → pr, pϕ[t] → pϕ};
```

We define the Newtonian Hamiltonian (in reduced variables):

In[4]:=
```
H[r_, pr_, pϕ_] := pr^2 / 2 + pϕ^2 / (2 r^2) - 1 / r
```

Then Hamilton’s equations are:

In[5]:=
```
dotϕ = D[H[r, pr, pϕ], pϕ]
```

Out[5]= $\frac{p\phi}{r^2}$

In[6]:=
```
dotr = D[H[r, pr, pϕ], pr]
```

Out[6]= pr

In[7]:=
```
dotpr = - D[H[r, pr, pϕ], r]
```

Out[7]= $\frac{p\phi^2}{r^3} - \frac{1}{r^2}$

In[8]:=
```
substituteEOMs = {r'[t] → dotr, ϕ'[t] → dotϕ, pr'[t] → dotpr, pϕ'[t] → 0};
```

We compute the quadrupole moment and its time derivatives

In[9]:=
```
M = r^2 {{Cos[ϕ]^2, Sin[ϕ] Cos[ϕ], 0}, {Sin[ϕ] Cos[ϕ], Sin[ϕ]^2, 0}, {0, 0, 1}};
Q = M - 1 / 3 IdentityMatrix[3] × Tr[M] // Simplify
```

Out[10]=

$$\left\{\left\{\frac{1}{6} r^2 (-1 + 3 \mathrm{Cos}[2 \phi]),\ r^2 \mathrm{Cos}[\phi] \mathrm{Sin}[\phi],\ 0\right\},\ \left\{r^2 \mathrm{Cos}[\phi] \mathrm{Sin}[\phi],\ -\frac{1}{6} r^2 (1 + 3 \mathrm{Cos}[2 \phi]),\ 0\right\},\ \left\{0,\ 0,\ \frac{r^2}{3}\right\}\right\}$$

In[11]:=
```
dQdt = D[Q /. addtdependence, t] /. removetdependence /. substituteEOMs //
  FullSimplify
```

Out[11]=

$$\left\{\left\{-\frac{pr\,r}{3}+pr\,r\,\text{Cos}[2\,\phi]-p\phi\,\text{Sin}[2\,\phi],\ p\phi\,\text{Cos}[2\,\phi]+pr\,r\,\text{Sin}[2\,\phi],\ 0\right\},\left\{p\phi\,\text{Cos}[2\,\phi]+pr\,r\,\text{Sin}[2\,\phi],\ -\frac{1}{3}\,pr\,r\,(1+3\,\text{Cos}[2\,\phi])+p\phi\,\text{Sin}[2\,\phi],\ 0\right\},\left\{0,\ 0,\ \frac{2\,pr\,r}{3}\right\}\right\}$$

In[12]:=
```
d2Qdt2 =
 D[dQdt /. addtdependence, t] /. removetdependence /. substituteEOMs //
  FullSimplify
```

Out[12]=

$$\left\{\left\{-\frac{1}{3\,r^2}\left(p\phi^2+r\left(-1+pr^2\,r\right)+3\left(p\phi^2+r-pr^2\,r^2\right)\text{Cos}[2\,\phi]+6\,pr\,p\phi\,r\,\text{Sin}[2\,\phi]\right),\ \frac{2\,pr\,p\phi\,r\,\text{Cos}[2\,\phi]-\left(p\phi^2+r-pr^2\,r^2\right)\text{Sin}[2\,\phi]}{r^2},\ 0\right\},\left\{\frac{2\,pr\,p\phi\,r\,\text{Cos}[2\,\phi]-\left(p\phi^2+r-pr^2\,r^2\right)\text{Sin}[2\,\phi]}{r^2},\ \frac{-p\phi^2+r-pr^2\,r^2+3\left(p\phi^2+r-pr^2\,r^2\right)\text{Cos}[2\,\phi]+6\,pr\,p\phi\,r\,\text{Sin}[2\,\phi]}{3\,r^2},\ 0\right\},\left\{0,\ 0,\ \frac{2\left(p\phi^2+r\left(-1+pr^2\,r\right)\right)}{3\,r^2}\right\}\right\}$$

In[13]:=
```
d3Qdt3 =
 D[d2Qdt2 /. addtdependence, t] /. removetdependence /. substituteEOMs //
  FullSimplify
```

Out[13]=

$$\left\{\left\{\frac{pr\,r-3\,pr\,r\,\text{Cos}[2\,\phi]+12\,p\phi\,\text{Sin}[2\,\phi]}{3\,r^3},\ -\frac{4\,p\phi\,\text{Cos}[2\,\phi]+pr\,r\,\text{Sin}[2\,\phi]}{r^3},\ 0\right\},\left\{-\frac{4\,p\phi\,\text{Cos}[2\,\phi]+pr\,r\,\text{Sin}[2\,\phi]}{r^3},\ \frac{pr\,r+3\,pr\,r\,\text{Cos}[2\,\phi]-12\,p\phi\,\text{Sin}[2\,\phi]}{3\,r^3},\ 0\right\},\left\{0,\ 0,\ -\frac{2\,pr}{3\,r^2}\right\}\right\}$$

Now, we compute the instantaneous energy flux:
1/5 d^3 Q_ij/dt^3 d^3 Q_ij/dt^3 :

In[14]:=
```
ΦE = 1 / 5 Sum[d3Qdt3[[i, j]] × d3Qdt3[[i, j]], {i, 1, 3}, {j, 1, 3}] // FullSimplify
```

Out[14]=

$$\frac{8\left(12\,p\phi^2+pr^2\,r^2\right)}{15\,r^6}$$

And we compute the instantaneous angular momentum flux:
2/5 ϵ^3ij d^2 Q_ij/dt^3 d^3 Q_ij/dt^3 :

In[15]:=
```
ΦL = 2 / 5 (Sum[d2Qdt2[[1, i]] × d3Qdt3[[2, i]], {i, 1, 3}] -
     Sum[d2Qdt2[[2, i]] × d3Qdt3[[1, i]], {i, 1, 3}]) // FullSimplify
```

Out[15]=

$$\frac{8\,p\phi\,\left(2\,p\phi^2 + r\,\left(2 - pr^2\,r\right)\right)}{5\,r^5}$$

Now, we move on to compute the orbit-average.
First we need the Keplerian transformations:

In[16]:=
```
toKeplerian = {r → a (1 - e^2) / (1 + e Cos[ϕ]),
   pr → e Sin[ϕ] / Sqrt[a (1 - e^2)], pϕ → Sqrt[a (1 - e^2)]};
```

Next, we obtain the period of the orbit:

In[17]:=
```
period = Integrate[1 / dotϕ /. toKeplerian // FullSimplify, {ϕ, 0, 2 Pi}]
```

Out[17]=

$$2\,a^{3/2}\,\pi$$

In this way, we compute the orbit-averaged energy:

In[18]:=
```
1 / period Integrate[ΦE / dotϕ /. toKeplerian // FullSimplify, {ϕ, 0, 2 Pi}] /.
  Sqrt[a - a e^2] → Sqrt[a] Sqrt[1 - e^2] // FullSimplify
```

Out[18]=

$$\frac{96 + 292\,e^2 + 37\,e^4}{15\,a^5\,\left(1 - e^2\right)^{7/2}}$$

And the orbit-averaged angular momentum:

In[19]:=
```
1 / period Integrate[ΦL / dotϕ /. toKeplerian // FullSimplify, {ϕ, 0, 2 Pi}] /.
  Sqrt[a - a e^2] → Sqrt[a] Sqrt[1 - e^2] // FullSimplify
```

Out[19]=

$$\frac{4\,\left(8 + 7\,e^2\right)}{5\,a^{7/2}\,\left(-1 + e^2\right)^2}$$

2d)

We define the Newtonian Hamiltonian (in reduced variables):

In[20]:=
```
H[r_, pr_, pϕ_] := pr^2 / 2 + pϕ^2 / (2 r^2) - 1 / r
```

Then Hamilton's equations are:

In[21]:=
```
dotϕ = D[H[r, pr, pϕ], pϕ]
```

Out[21]=

$$\frac{p\phi}{r^2}$$

In[22]:=
```
dotr = D[H[r, pr, pϕ], pr]
```

Out[22]=

pr

In[23]:=
```
dotpr = - D[H[r, pr, pϕ], r]
```
Out[23]=

$$\frac{p\phi^2}{r^3} - \frac{1}{r^2}$$

We define the RR force components:

In[24]:=
```
Fr[r_, pr_, pϕ_] := 8 ν pr / (15 r^3)
    ((-3 α + 9 β + 3) (pr^2 + pϕ^2 / r^2) + (9 α - 15 β + 9) pr^2 + (9 α - 9 β + 17) / r);
Fϕ[r_, pr_, pϕ_] :=
  8 ν pϕ / (15 r^3) (9 (α + 1) pr^2 - 3 (2 + α) (pr^2 + pϕ^2 / r^2) + 3 (α - 2) / r);
```

The transformation to Keplerian parametrization is:

In[26]:=
```
toKeplerian = {r → a (1 - e^2) / (1 + e Cos[ϕ]),
    pr → e Sin[ϕ] / Sqrt[a (1 - e^2)], pϕ → Sqrt[a (1 - e^2)]};
```

And we compute the following quantities in the Keplerian parametrization:

In[27]:=
```
integrandE = dotr Fr[r, pr, pϕ] + dotϕ Fϕ[r, pr, pϕ] /. toKeplerian // FullSimplify
```
Out[27]=

$$\frac{1}{15\,a^5\,(-1+e^2)^5}\,\nu\,(1+e\,\mathrm{Cos}[\phi])^3\,\Big(96-4\,e^2\,\big(-31+3\,\alpha+3\,e^2\,(2+\alpha)\big)+9\,e^4\,\beta+2\,e\,\big(12\,(14+\alpha)+e^2\,(61+12\,\alpha-9\,\beta)\big)\,\mathrm{Cos}[\phi]+4\,e^2\,\big(77+21\,\alpha+3\,e^2\,(6+3\,\alpha-2\,\beta)\big)\,\mathrm{Cos}[2\,\phi]+e^3\,(2\,(59+24\,\alpha+9\,\beta)\,\mathrm{Cos}[3\,\phi]+15\,e\,\beta\,\mathrm{Cos}[4\,\phi])\Big)$$

In[28]:=
```
integrandL = Fϕ[r, pr, pϕ] /. toKeplerian // FullSimplify
```
Out[28]=

$$-\frac{1}{5\,a^4\,(-1+e^2)^4}\,4\,\sqrt{a-a\,e^2}\,\nu\,(1+e\,\mathrm{Cos}[\phi])^3\,\big(8+e^2-e^2\,\alpha+2\,e\,(6+\alpha)\,\mathrm{Cos}[\phi]+3\,e^2\,(1+\alpha)\,\mathrm{Cos}[2\,\phi]\big)$$

Now, we perform the orbit-averages to obtain the fluxes:

In[29]:=
```
period = Integrate[1 / dotϕ /. toKeplerian // FullSimplify, {ϕ, 0, 2 Pi}]
```
Out[29]=

$$2\,a^{3/2}\,\pi$$

In[30]:=
```
orbAvL =
  - 1 / period Integrate[integrandL / dotϕ  /. toKeplerian // FullSimplify,
     {ϕ, 0, 2 Pi}] // FullSimplify
```
Out[30]=

$$\frac{4\,(8+7\,e^2)\,\nu}{5\,a^{7/2}\,(-1+e^2)^2}$$

In[31]:=
```
orbAvE =
  - 1 / period Integrate[integrandE / dotϕ  /. toKeplerian // FullSimplify,
     {ϕ, 0, 2 Pi}] // FullSimplify
```
Out[31]=

$$\frac{\left(96 + 292\, e^2 + 37\, e^4\right) \nu}{15\, a^5 \left(1 - e^2\right)^{7/2}}$$

2 e)

With the code below, we can play with the parameters to get interesting trajectories

In[32]:=
```
addtdependence = {ϕ → ϕ[t], r → r[t], pr → pr[t], pϕ → pϕ[t]};
numvalues = {α → -16 / 3, β → -13 / 2, ν → 0.25};
initvalues = {e → 0.3, ϕ → 0, R → 20};
```
In[35]:=
```
r0 = R / (1 + e Cos[ϕ]) /. initvalues
pr0 = e Sin[ϕ] / Sqrt[R] /. initvalues
pϕ0 = Sqrt[R] /. initvalues
```
Out[35]=
```
15.3846
```
Out[36]=
```
0.
```
Out[37]=

$2 \sqrt{5}$

In[38]:=

```
tfinal = 12 960; (* This time is selected by exploring larger times,
and realizing that the numerical solver breaks at some point,
so we choose some time before it breaks *)
s = NDSolve[
  {
   r'[t] == (D[H[r, pr, pϕ], pr] /. addtdependence /. numvalues),
   ϕ'[t] == (D[H[r, pr, pϕ], pϕ] /. addtdependence /. numvalues),
   pr'[t] ==
    (-D[H[r, pr, pϕ], r] + Fr[r, pr, pϕ] /. addtdependence /. numvalues),
   pϕ'[t] == (Fϕ[r, pr, pϕ] /. addtdependence /. numvalues),
   r[0] == r0,
   ϕ[0] == 0,
   pr[0] == pr0,
   pϕ[0] == pϕ0
  },
  {r, ϕ, pr, pϕ},
  {t, tfinal}
 ]
```

Out[39]=

```
{{r → InterpolatingFunction[ Domain: {{0., 13 000.}} Output: scalar ],
  ϕ → InterpolatingFunction[ Domain: {{0., 13 000.}} Output: scalar ],
  pr → InterpolatingFunction[ Domain: {{0., 13 000.}} Output: scalar ],
  pϕ → InterpolatingFunction[ Domain: {{0., 13 000.}} Output: scalar ]}}
```

In[40]:=
```
ParametricPlot[Evaluate[{r[t] Cos[ϕ[t]], r[t] Sin[ϕ[t]]} /. s], {t, 0, tfinal}]
```

Out[40]=